

NAG Library Routine Document

G01EEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G01EEF computes the upper and lower tail probabilities and the probability density function of the beta distribution with parameters a and b .

2 Specification

```
SUBROUTINE G01EEF (X, A, B, TOL, P, Q, PDF, IFAIL)
```

```
INTEGER IFAIL
```

```
REAL (KIND=nag_wp) X, A, B, TOL, P, Q, PDF
```

3 Description

The probability density function of the beta distribution with parameters a and b is:

$$f(B : a, b) = \frac{\Gamma(a+b)}{\Gamma(a)\Gamma(b)} B^{a-1} (1-B)^{b-1}, \quad 0 \leq B \leq 1; a, b > 0.$$

The lower tail probability, $P(B \leq \beta : a, b)$ is defined by

$$P(B \leq \beta : a, b) = \frac{\Gamma(a+b)}{\Gamma(a)\Gamma(b)} \int_0^\beta B^{a-1} (1-B)^{b-1} dB = I_\beta(a, b), \quad 0 \leq \beta \leq 1; a, b > 0.$$

The function $I_x(a, b)$, also known as the incomplete beta function is calculated using S14CCF.

4 References

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

- | | | |
|----|--|--------------|
| 1: | X – REAL (KIND=nag_wp)
<i>On entry:</i> β , the value of the beta variate.
<i>Constraint:</i> $0.0 \leq X \leq 1.0$. | <i>Input</i> |
| 2: | A – REAL (KIND=nag_wp)
<i>On entry:</i> a , the first parameter of the required beta distribution.
<i>Constraint:</i> $0.0 < A \leq 10^6$. | <i>Input</i> |
| 3: | B – REAL (KIND=nag_wp)
<i>On entry:</i> b , the second parameter of the required beta distribution.
<i>Constraint:</i> $0.0 < B \leq 10^6$. | <i>Input</i> |
| 4: | TOL – REAL (KIND=nag_wp)
<i>On entry:</i> this parameter is no longer referenced, but is included for backwards compatibility. | <i>Input</i> |

- 5: P – REAL (KIND=nag_wp) Output
On exit: the lower tail probability, $P(B \leq \beta : a, b)$.
- 6: Q – REAL (KIND=nag_wp) Output
On exit: the upper tail probability, $P(B \geq \beta : a, b)$.
- 7: PDF – REAL (KIND=nag_wp) Output
On exit: the probability density function, $f(B : a, b)$.
- 8: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
- For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL \neq 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
- On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01EEF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $X < 0.0$,
 or $X > 1.0$.

IFAIL = 2

On entry, $A \leq 0.0$,
 or $A > 10^6$,
 or $B \leq 0.0$,
 or $B > 10^6$.

IFAIL = 4

X is too far out into the tails for the probability to be evaluated exactly. The results returned are 0 and 1 as appropriate. These should be a good approximation to the required solution.

7 Accuracy

The accuracy is limited by the error in the incomplete beta function. See Section 7 in S14CCF for further details.

8 Further Comments

None.

9 Example

This example reads values from a number of beta distributions and computes the associated upper and lower tail probabilities and the corresponding value of the probability density function.

9.1 Program Text

```

Program g01eefe

!      G01EEF Example Program Text
!
!      Mark 24 Release. NAG Copyright 2012.
!
!      .. Use Statements ..
Use nag_library, Only: g01eef, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: a, b, p, pdf, q, tol, x
Integer                    :: ifail
!      .. Executable Statements ..
Write (nout,*) 'G01EEF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Display titles
Write (nout,*) '      X      A      B      P      Q      PDF'
Write (nout,*)

d_lp: Do
  Read (nin,*,Iostat=ifail) x, a, b
  If (ifail/=0) Then
    Exit d_lp
  End If

!      Calculate probability
!      NB: parameter tol is no longer referenced
  ifail = -1
  Call g01eef(x,a,b,tol,p,q,pdf,ifail)
  If (ifail/=0) Then
    If (ifail/=3 .And. ifail/=4) Then
      Exit d_lp
    End If
  End If

!      Display results
  Write (nout,99999) x, a, b, p, q, pdf
End Do d_lp

99999 Format (1X,6(F7.4,2X),A,I1)
End Program g01eefe

```

9.2 Program Data

```

G01EEF Example Program Data
0.25 1.0 2.0
0.75 1.5 1.5
0.5 2.0 1.0

```

9.3 Program Results

G01EEF Example Program Results

X	A	B	P	Q	PDF
0.2500	1.0000	2.0000	0.4375	0.5625	1.5000
0.7500	1.5000	1.5000	0.8045	0.1955	1.1027
0.5000	2.0000	1.0000	0.2500	0.7500	1.0000
